version 17.0
clear
set more off

cd "INSERT PATH TO WORKING DIRECTORY HERE"

use "PULSE_complete.dta", clear

*******************************************************************************
***				MULTIPLY IMPUTATION MODELS FOR EIPRECEIPT					***
*******************************************************************************

*** MULTIPLE IMPUTATIONS: https://stats.idre.ucla.edu/stata/seminars/mi_in_stata_pt1_new/ ***

keep if WEEK>=22 & WEEK<=33
keep HHID SOURCESPEND4 AGE FEMALE EIPRECEIPT HHOUSES INCCAT2 RACEETH2 EDUC fam_status LOSTJOB ANYWORK WEEK month PWEIGHT EST_ST EIP_YN MARRIED THHLD_NUMPER WORKING MENTALHLTH2 DISABILITY2 EIP EIPSAVE
save "EIPRECEIPT_MI.dta", replace 

mdesc HHID SOURCESPEND4 AGE FEMALE EIPRECEIPT HHOUSES INCCAT2 RACEETH2 EDUC fam_status LOSTJOB ANYWORK WEEK month PWEIGHT EST_ST EIP_YN MARRIED THHLD_NUMPER WORKING MENTALHLTH2 DISABILITY2 if EIP<. 

mdesc HHID SOURCESPEND4 AGE FEMALE EIPRECEIPT HHOUSES INCCAT2 RACEETH2 EDUC fam_status LOSTJOB ANYWORK WEEK month PWEIGHT EST_ST EIP_YN MARRIED THHLD_NUMPER WORKING MENTALHLTH2 DISABILITY2 if EIP_YN<. 

mi set flong
g agesquare=AGE^2
mi register imputed SOURCESPEND4 HHOUSES EIPRECEIPT INCCAT2 MENTALHLTH2 EIPSAVE
mi impute chained (ologit) SOURCESPEND4 HHOUSES EIPRECEIPT MENTALHLTH2 INCCAT2 EIPSAVE = AGE agesquare FEMALE i.EDUC THHLD_NUMPER i.MARRIED WORKING DISABILITY2 i.EST_ST, add(5) rseed(53421) noimputed force by(RACEETH2)

save "MI_EIPRECEIPT.dta", replace 

*** 		CREATES AVERAGE BY WEEEK FOR EACH OBS, EXCLUDING CURRENT STATE	***

bys WEEK RACEETH2: asrol EIPRECEIPT if EIPRECEIPT<., stat(mean) xf(EST_ST) gen(EIPRECEIPT_IV)

mi svyset [pw=PWEIGHT], strata(WEEK)

eststo clear

foreach race in 1 2 3 4 5 {

display "RACEETH2=`race'"

mi estimate, or post esampvaryok: svy: logit SOURCESPEND4 AGE agesquare FEMALE i.EDUC i.INCCAT2 THHLD_NUMPER i.MARRIED WORKING MENTALHLTH2 DISABILITY2 HHOUSES EIPRECEIPT i.month i.EST_ST if WEEK>22 & RACEETH2==`race'
eststo
mi estimate, or post esampvaryok: svy: logit SOURCESPEND4 AGE agesquare FEMALE i.EDUC i.INCCAT2 THHLD_NUMPER i.MARRIED WORKING MENTALHLTH2 DISABILITY2 EIPRECEIPT##HHOUSES i.month i.EST_ST if WEEK>22 & RACEETH2==`race'
eststo

}

esttab using MI_EIPRECEIPT_INFORMFINSUPPORT.csv, replace star(* 0.10 ** 0.05 *** 0.01) cells(b(star fmt(4)) se(fmt(4) par)) label legend varlabels(_cons Constant) mtitles(EIP_1 EIP_1_IA EIP_2 EIP_2_IA EIP_3 EIP_3_IA EIP_4 EIP_4_IA EIP_5 EIP_5_IA) style(fixed) unstack eform
eststo clear

*** 	TESTING FOR ENDOGENEITY USING ivregress; ivprobit ONLY WORKS WITH 	***
*** 	CONTINUOUS REGRESSORS AND biprobit GENERATES ERROR MESSAGE AND 		***
***		TAKES TOO LONG														***

foreach race in 1 2 3 4 5 {

display "RACEETH2=`race'"

mi estimate, post esampvaryok: regress SOURCESPEND4 AGE agesquare FEMALE i.EDUC i.INCCAT2 THHLD_NUMPER i.MARRIED WORKING MENTALHLTH2 DISABILITY2 HHOUSES EIPRECEIPT ib13.month i.EST_ST [pw=PWEIGHT] if WEEK>22 & RACEETH2==`race'
eststo
mi estimate, esampvaryok cmdok post: ivregress 2sls SOURCESPEND4 AGE agesquare FEMALE i.EDUC i.INCCAT2 THHLD_NUMPER i.MARRIED WORKING MENTALHLTH2 DISABILITY2 HHOUSES ib13.month i.EST_ST (EIPRECEIPT = EIPRECEIPT_IV) [pw=PWEIGHT] if WEEK>22 & RACEETH2==`race', first
eststo

}

esttab using MI_EIPRECEIPT_INFORMFINSUPPORT_IV.csv, replace star(* 0.10 ** 0.05 *** 0.01) cells(b(star fmt(4)) se(fmt(4) par)) label legend varlabels(_cons Constant) mtitles(EIP_1 EIP_1_IV EIP_2 EIP_2_IV EIP_3 EIP_3_IV EIP_4 EIP_4_IV EIP_5 EIP_5_IV) style(fixed) unstack
eststo clear

*******************************************************************************
*** 		MULTIPLE IMPUTATION MODELS FOR CTCRECEIPT						***
*******************************************************************************

use "PULSE_complete.dta", clear

*** MULTIPLE IMPUTATIONS: https://stats.idre.ucla.edu/stata/seminars/mi_in_stata_pt1_new/ ***

keep if WEEK>=34 & WEEK<=42 & PARENT==1
keep HHID SOURCESPEND4 AGE FEMALE CTCRECEIPT HHOUSES INCCAT2 RACEETH2 EDUC fam_status LOSTJOB ANYWORK WEEK month PWEIGHT EST_ST PARENT CTC_YN VACCINESTATUS THHLD_NUMPER MARRIED WORKING MENTALHLTH2 DISABILITY2 NOPUBSCHOOL CTCSAVE REGION
save "CTCRECEIPT_MI.dta", replace

mi set flong
g agesquare=AGE^2
mdesc HHID SOURCESPEND4 AGE FEMALE CTCRECEIPT HHOUSES INCCAT2 RACEETH2 EDUC fam_status LOSTJOB ANYWORK WEEK month PWEIGHT EST_ST PARENT CTC_YN VACCINESTATUS SOURCESPEND4 AGE THHLD_NUMPER MARRIED WORKING MENTALHLTH2 DISABILITY2
mi register imputed SOURCESPEND4 HHOUSES CTCRECEIPT INCCAT2 MENTALHLTH2 CTCSAVE
mi impute chained (ologit) SOURCESPEND4 CTCRECEIPT HHOUSES MENTALHLTH2 INCCAT2 = AGE agesquare FEMALE i.EDUC THHLD_NUMPER i.MARRIED WORKING DISABILITY2 i.EST_ST, add(5) rseed(53421) noimputed force by(RACEETH2)

save "MI_CTCRECEIPT.dta", replace

bys WEEK: asrol CTCRECEIPT if CTCRECEIPT<., stat(mean) xf(EST_ST) gen(CTCRECEIPT_IV)

mi svyset [pw=PWEIGHT], strata(WEEK)

eststo clear

foreach race in 1 2 3 4 5 {

display "RACEETH2=`race'"

mi estimate, or post esampvaryok: svy: logit SOURCESPEND4 AGE agesquare FEMALE ib1.EDUC ib1.INCCAT2 THHLD_NUMPER ib1.MARRIED WORKING MENTALHLTH2 DISABILITY2 HHOUSES CTCRECEIPT ib19.month ib6.EST_ST if PARENT==1 & RACEETH2==`race'
eststo
mi estimate, or post esampvaryok: svy: logit SOURCESPEND4 AGE agesquare FEMALE ib1.EDUC ib1.INCCAT2 THHLD_NUMPER ib1.MARRIED WORKING MENTALHLTH2 DISABILITY2 CTCRECEIPT##HHOUSES ib19.month ib6.EST_ST if PARENT==1 & RACEETH2==`race'
eststo

}

esttab using MI_CTCRECEIPT_INFORMFINSUPPORT.csv, replace star(* 0.10 ** 0.05 *** 0.01) cells(b(star fmt(4)) se(fmt(4) par)) label legend varlabels(_cons Constant) mtitles(CTC_1 CTC_1_IA CTC_2 CTC_2_IA CTC_3 CTC_3_IA CTC_4 CTC_4_IA CTC_5 CTC_5_IA) style(fixed) unstack eform

eststo clear

foreach race in 1 2 3 4 5 {

display "RACEETH2=`race'"

mi estimate, post esampvaryok imputations(1 2 3 4 5): regress SOURCESPEND4 AGE agesquare FEMALE ib1.EDUC ib1.INCCAT2 THHLD_NUMPER ib1.MARRIED WORKING MENTALHLTH2 DISABILITY2 HHOUSES CTCRECEIPT ib19.month ib6.EST_ST [pw=PWEIGHT] if WEEK>22 & PARENT==1 & RACEETH2==`race'
eststo
mi estimate, cmdok post esampvaryok imputations(1 2 3 4 5): ivregress 2sls SOURCESPEND4 AGE agesquare FEMALE ib1.EDUC ib1.INCCAT2 THHLD_NUMPER ib1.MARRIED WORKING MENTALHLTH2 DISABILITY2 HHOUSES ib19.month ib6.EST_ST (CTCRECEIPT = CTCRECEIPT_IV) [pw=PWEIGHT] if WEEK>22 & PARENT==1 & RACEETH2==`race', first
eststo
mi estimate, cmdok post esampvaryok imputations(1 2 3 4 5): ivregress 2sls SOURCESPEND4 AGE agesquare FEMALE ib1.EDUC ib1.INCCAT2 THHLD_NUMPER ib1.MARRIED WORKING MENTALHLTH2 DISABILITY2 HHOUSES ib19.month ib6.EST_ST (CTCRECEIPT = NOPUBSCHOOL) [pw=PWEIGHT] if WEEK>22 & PARENT==1 & RACEETH2==`race', first
eststo

}

esttab using MI_CTCRECEIPT_INFORMFINSUPPORT_IV.csv, replace star(* 0.10 ** 0.05 *** 0.01) cells(b(star fmt(4)) se(fmt(4) par)) label legend varlabels(_cons Constant) mtitles(CTC_1 CTC_1_IV CTC_1_PUB CTC_2 CTC_2_IV CTC_2_PUB CTC_3 CTC_3_IV CTC_3_PUB CTC_4 CTC_4_IV CTC_4_PUB CTC_5 CTC_5_IV CTC_5_PUB) style(fixed) unstack

eststo clear

*******************************************************************************
***			MULTIPLE IMPUTATION MODELS FOR UIRECEIPT						***
*******************************************************************************

use "PULSE_complete.dta", clear

*** MULTIPLE IMPUTATIONS: https://stats.idre.ucla.edu/stata/seminars/mi_in_stata_pt1_new/ ***

keep if inlist(WEEK, 28 29, 30, 31, 32, 33)
keep HHID SOURCESPEND4 AGE FEMALE UIRECEIPT UI_RECVRV HHOUSES INCCAT2 RACEETH2 EDUC fam_status LOSTJOB ANYWORK WORKING WEEK month PWEIGHT EST_ST PARENT RSNNOTWORK7 RSNNOTWORK1 THHLD_NUMPER MARRIED WORKING MENTALHLTH2 DISABILITY2 VACCINESTATUS REGION

save "UIRECEIPT_MI.dta", replace

mi set flong
g agesquare=AGE^2
mi register imputed SOURCESPEND4 HHOUSES UIRECEIPT MENTALHLTH2 INCCAT2 VACCINESTATUS
mi impute chained (ologit) SOURCESPEND4 UIRECEIPT HHOUSES MENTALHLTH2 VACCINESTATUS INCCAT2 = AGE agesquare FEMALE i.EDUC THHLD_NUMPER i.MARRIED WORKING DISABILITY2 i.REGION, add(5) rseed(53421) noimputed force augment by(RACEETH2)

save "MI_UIRECEIPT.dta", replace

bys WEEK: asrol UIRECEIPT if UIRECEIPT<., stat(mean) xf(EST_ST) gen(UIRECEIPT_IV)

mi svyset [pw=PWEIGHT], strata(WEEK)

foreach race in 1 2 3 4 5 {

display "RACEETH2=`race'"

mi estimate, or post esampvaryok: svy: logit SOURCESPEND4 AGE agesquare FEMALE i.EDUC i.INCCAT2 THHLD_NUMPER i.MARRIED WORKING MENTALHLTH2 DISABILITY2 HHOUSES UIRECEIPT i.month i.REGION if inlist(WEEK, 28, 29, 30, 31, 32, 33) & WORKING==0 & RSNNOTWORK1==0 & RSNNOTWORK7==0 & RACEETH2==`race', or
eststo
mi estimate, or post esampvaryok: svy: logit SOURCESPEND4 AGE agesquare FEMALE i.EDUC i.INCCAT2 THHLD_NUMPER i.MARRIED WORKING MENTALHLTH2 DISABILITY2 UIRECEIPT##HHOUSES i.month i.REGION if inlist(WEEK, 28, 29, 30, 31, 32, 33) & WORKING==0 & RSNNOTWORK1==0 & RSNNOTWORK7==0 & RACEETH2==`race', or
eststo

}

esttab using MI_UIRECEIPT_INFORMFINSUPPORT.csv, replace star(* 0.10 ** 0.05 *** 0.01) cells(b(star fmt(4)) se(fmt(4) par)) label legend varlabels(_cons Constant) mtitles(UI_1 UI_1_IA UI_2 UI_2_IA UI_3 UI_3_IA UI_4 UI_4_IA UI_5 UI_5_IA) style(fixed) unstack eform

eststo clear

foreach race in 1 2 3 4 5 {

display "RACEETH2=`race'"

mi estimate, post esampvaryok: regress SOURCESPEND4 AGE agesquare FEMALE i.EDUC i.INCCAT2 THHLD_NUMPER i.MARRIED WORKING MENTALHLTH2 DISABILITY2 HHOUSES UIRECEIPT i.month i.REGION [pw=PWEIGHT] if inlist(WEEK, 28, 29, 30, 31, 32, 33) & WORKING==0 & RSNNOTWORK1==0 & RSNNOTWORK7==0 & RACEETH2==`race'
eststo
mi estimate, post cmdok esampvaryok: ivregress 2sls SOURCESPEND4 AGE agesquare FEMALE HHOUSES i.EDUC i.INCCAT2 THHLD_NUMPER i.MARRIED WORKING MENTALHLTH2 DISABILITY2 i.month i.REGION (UIRECEIPT = UIRECEIPT_IV) [pw=PWEIGHT] if inlist(WEEK, 28, 29, 30, 31, 32, 33) & WORKING==0 & RSNNOTWORK1==0 & RSNNOTWORK7==0 & RACEETH2==`race', first
eststo
mi estimate, post cmdok esampvaryok: ivregress 2sls SOURCESPEND4 AGE agesquare FEMALE HHOUSES i.EDUC i.INCCAT2 THHLD_NUMPER i.MARRIED WORKING MENTALHLTH2 DISABILITY2 i.month i.REGION (UIRECEIPT = VACCINESTATUS) [pw=PWEIGHT] if inlist(WEEK, 28, 29, 30, 31, 32, 33) & WORKING==0 & RSNNOTWORK1==0 & RSNNOTWORK7==0 & RACEETH2==`race', first
eststo
}

esttab using UIRECEIPT_INFORMFINSUPPORT_IV.csv, replace star(* 0.10 ** 0.05 *** 0.01) cells(b(star fmt(4))) se(fmt(4)) par label legend varlabels(_cons Constant) mtitles(UI_1 UI_1_IV UI_1_VACC UI_2 UI_2_IV UI_2_VACC UI_3 UI_3_IV UI_3_VACC UI_4 UI_4_IV UI_4_VACC UI_5 UI_5_IV UI_5_VACC) style(fixed) unstack

eststo clear

*******************************************************************************
***		ALTERNATIVE MODEL FOR CTC RECEIPT WITH PROXY FOR LIQUID ASSETS		***
*******************************************************************************

use "PULSE_complete.dta", clear

keep if WEEK>=22 & WEEK<=33
keep HHID SOURCESPEND4 AGE FEMALE EIPRECEIPT HHOUSES INCCAT2 RACEETH2 EDUC fam_status LOSTJOB ANYWORK WEEK month PWEIGHT EST_ST EIP_YN MARRIED THHLD_NUMPER WORKING MENTALHLTH2 DISABILITY2 EIP EIPSAVE REGION
save "EIPRECEIPT_MI.dta", replace 

mdesc HHID SOURCESPEND4 AGE FEMALE EIPRECEIPT HHOUSES INCCAT2 RACEETH2 EDUC fam_status LOSTJOB ANYWORK WEEK month PWEIGHT EST_ST EIP_YN MARRIED THHLD_NUMPER WORKING MENTALHLTH2 DISABILITY2 if EIP<. 

mdesc HHID SOURCESPEND4 AGE FEMALE EIPRECEIPT HHOUSES INCCAT2 RACEETH2 EDUC fam_status LOSTJOB ANYWORK WEEK month PWEIGHT EST_ST EIP_YN MARRIED THHLD_NUMPER WORKING MENTALHLTH2 DISABILITY2 if EIP_YN<. 

mi set flong
g agesquare=AGE^2
mi register imputed SOURCESPEND4 HHOUSES EIPRECEIPT INCCAT2 MENTALHLTH2 EIPSAVE
mi impute chained (ologit) SOURCESPEND4 HHOUSES EIPRECEIPT MENTALHLTH2 INCCAT2 EIPSAVE = AGE agesquare FEMALE i.EDUC THHLD_NUMPER i.MARRIED WORKING DISABILITY2 i.REGION, add(5) rseed(53421) noimputed force by(RACEETH2)

save "MI_EIPRECEIPT_SAVE.dta", replace 

*** 	ESTIMATING THE CORRELATION BETWEEN EIPSAVE AND SOURCESPEND4	***

mi svyset [pw=PWEIGHT], strata(WEEK)

foreach race in 1 2 3 4 5 {

display "RACEETH2=`race'"

mi estimate, post esampvaryok: svy: logit SOURCESPEND4 AGE agesquare FEMALE i.EDUC i.INCCAT2 HHOUSES EIPSAVE i.month i.REGION if WEEK>22 & RACEETH2==`race' & EIPRECEIPT==1, or
eststo
}

esttab using EIPSAVE_INFORMFINSUPPORT.csv, replace star(* 0.10 ** 0.05 *** 0.01) cells(b(star fmt(4)) se(fmt(4) par)) label legend varlabels(_cons Constant) mtitles(EIPSAVE_1 EIPSAVE_2 EIPSAVE_3 EIPSAVE_4 EIPSAVE_5) style(fixed) unstack eform
eststo clear

*** 						CTC SAVE								***

use "PULSE_complete.dta", clear

*** MULTIPLE IMPUTATIONS: https://stats.idre.ucla.edu/stata/seminars/mi_in_stata_pt1_new/ ***

keep if WEEK>=34 & WEEK<=42 & PARENT==1
keep HHID SOURCESPEND4 AGE FEMALE CTCRECEIPT HHOUSES INCCAT2 RACEETH2 EDUC fam_status LOSTJOB ANYWORK WEEK month PWEIGHT EST_ST PARENT CTC_YN VACCINESTATUS THHLD_NUMPER MARRIED WORKING MENTALHLTH2 DISABILITY2 NOPUBSCHOOL CTCSAVE REGION
save "CTCRECEIPT_MI.dta", replace

mi set flong
g agesquare=AGE^2
mdesc HHID SOURCESPEND4 AGE FEMALE CTCRECEIPT HHOUSES INCCAT2 RACEETH2 EDUC fam_status LOSTJOB ANYWORK WEEK month PWEIGHT EST_ST PARENT CTC_YN VACCINESTATUS SOURCESPEND4 AGE THHLD_NUMPER MARRIED WORKING MENTALHLTH2 DISABILITY2
mi register imputed SOURCESPEND4 HHOUSES CTCRECEIPT INCCAT2 MENTALHLTH2 CTCSAVE
mi impute chained (ologit) SOURCESPEND4 CTCRECEIPT HHOUSES MENTALHLTH2 INCCAT2 CTCSAVE = AGE agesquare FEMALE i.EDUC THHLD_NUMPER i.MARRIED WORKING DISABILITY2 i.REGION, add(5) rseed(53421) noimputed force by(RACEETH2)

save "MI_CTCRECEIPT_SAVE.dta", replace

*** 	ESTIMATING THE CORRELATION BETWEEN CTCSAVE AND SOURCESPEND4	***
mi svyset [pw=PWEIGHT], strata(WEEK)


foreach race in 1 2 3 4 5 {

display "RACEETH2=`race'"

mi estimate, post esampvaryok: svy: logit SOURCESPEND4 AGE agesquare FEMALE i.EDUC i.INCCAT2 HHOUSES CTCSAVE i.month i.REGION if WEEK>22 & RACEETH2==`race' & CTCRECEIPT==1, or
eststo
}

esttab using CTCSAVE_INFORMFINSUPPORT.csv, replace star(* 0.10 ** 0.05 *** 0.01) cells(b(star fmt(4)) se(fmt(4) par)) label legend varlabels(_cons Constant) mtitles(CTCSAVE_1  CTCSAVE_2 CTCSAVE_3 CTCSAVE_4 CTCSAVE_5) style(fixed) unstack eform
eststo clear

*******************************************************************************
***		SUPPLEMENTAL MATERIAL: CTC CHECKS FOR LOW INCOME PARENTS			***
*******************************************************************************

use "MI_CTCRECEIPT.dta", clear

bys WEEK: asrol CTCRECEIPT if CTCRECEIPT<., stat(mean) xf(EST_ST) gen(CTCRECEIPT_IV)

mi svyset [pw=PWEIGHT], strata(WEEK)

eststo clear

foreach race in 1 2 3 4 5 {

display "RACEETH2=`race'"

mi estimate, or post esampvaryok: svy: logit SOURCESPEND4 AGE agesquare FEMALE ib1.EDUC ib1.INCCAT2 THHLD_NUMPER ib1.MARRIED WORKING MENTALHLTH2 DISABILITY2 HHOUSES CTCRECEIPT i.month i.REGION if PARENT==1 & RACEETH2==`race' & INCCAT2==1
eststo
mi estimate, or post esampvaryok: svy: logit SOURCESPEND4 AGE agesquare FEMALE ib1.EDUC ib1.INCCAT2 THHLD_NUMPER ib1.MARRIED WORKING MENTALHLTH2 DISABILITY2 CTCRECEIPT##HHOUSES i.month i.REGION if PARENT==1 & RACEETH2==`race' & INCCAT2==1
eststo

}

esttab using MI_CTCRECEIPT_INFORMFINSUPPORT_LI.csv, replace star(* 0.10 ** 0.05 *** 0.01) cells(b(star fmt(4)) se(fmt(4) par)) label legend varlabels(_cons Constant) mtitles(CTC_1 CTC_1_IA CTC_2 CTC_2_IA CTC_3 CTC_3_IA CTC_4 CTC_4_IA CTC_5 CTC_5_IA) style(fixed) unstack eform

eststo clear

foreach race in 1 2 3 4 5 {

display "RACEETH2=`race'"

mi estimate, post esampvaryok imputations(1 2 3 4 5): regress SOURCESPEND4 AGE agesquare FEMALE ib1.EDUC ib1.INCCAT2 THHLD_NUMPER ib1.MARRIED WORKING MENTALHLTH2 DISABILITY2 HHOUSES CTCRECEIPT i.month i.REGION [pw=PWEIGHT] if WEEK>22 & PARENT==1 & RACEETH2==`race' & INCCAT2==1
eststo
mi estimate, cmdok post esampvaryok imputations(1 2 3 4 5): ivregress 2sls SOURCESPEND4 AGE agesquare FEMALE ib1.EDUC ib1.INCCAT2 THHLD_NUMPER ib1.MARRIED WORKING MENTALHLTH2 DISABILITY2 HHOUSES i.month i.REGION (CTCRECEIPT = CTCRECEIPT_IV) [pw=PWEIGHT] if WEEK>22 & PARENT==1 & RACEETH2==`race' & INCCAT2==1, first
eststo

}

esttab using MI_CTCRECEIPT_INFORMFINSUPPORT_IV_LI.csv, replace star(* 0.10 ** 0.05 *** 0.01) cells(b(star fmt(4)) se(fmt(4) par)) label legend varlabels(_cons Constant) mtitles(CTC_1 CTC_1_IV CTC_2 CTC_2_IV CTC_3 CTC_3_IV CTC_4 CTC_4_IV CTC_5 CTC_5_IV) style(fixed) unstack

eststo clear







